/*
 * ESP32-C6 OLED + BMP280 显示程序
 * 显示内容：
 *   第一行：iCEasy
 *   第二、三、四行：温度、气压、海拔数据
 * 
 * 硬件连接：
 *   OLED & BMP280 共享I2C总线
 *   SDA -> GPIO6
 *   SCL -> GPIO7
 *   VCC -> 3.3V
 *   GND -> GND
 *   
 *   BMP280额外引脚：
 *   CSB -> 3.3V (选择I2C模式)
 *   SDO -> GND (I2C地址0x76) 或 3.3V (I2C地址0x77)
 */

#include <Wire.h>
#include <Adafruit_GFX.h>
#include <Adafruit_SSD1306.h>
#include <Adafruit_BMP280.h>

// OLED显示屏配置
#define SCREEN_WIDTH 128
#define SCREEN_HEIGHT 64
#define OLED_RESET -1
#define SCREEN_ADDRESS 0x3C

// I2C引脚定义（ESP32-C6）
#define I2C_SDA 6
#define I2C_SCL 7

// BMP280 I2C地址 (0x76或0x77，根据SDO接线)
#define BMP280_ADDRESS 0x76

// 创建对象
Adafruit_SSD1306 display(SCREEN_WIDTH, SCREEN_HEIGHT, &Wire, OLED_RESET);
Adafruit_BMP280 bmp;

// 海拔参考值（海平面气压，单位hPa）
#define SEALEVELPRESSURE_HPA (1013.25)

void setup() {
  Serial.begin(115200);
  delay(100);
  Serial.println("\n========================================");
  Serial.println("ESP32-C6 OLED + BMP280 Display System");
  Serial.println("========================================");
  
  // 初始化I2C总线
  Wire.begin(I2C_SDA, I2C_SCL);
  Serial.println("I2C initialized on SDA=GPIO6, SCL=GPIO7");
  
  // 初始化OLED显示屏
  if(!display.begin(SSD1306_SWITCHCAPVCC, SCREEN_ADDRESS)) {
    Serial.println("ERROR: SSD1306 allocation failed!");
    for(;;); // 死循环
  }
  Serial.println("OLED display initialized successfully");
  
  // 初始化BMP280传感器
  if (!bmp.begin(BMP280_ADDRESS)) {
    Serial.println("ERROR: Could not find BMP280 sensor!");
    Serial.println("Check wiring or try address 0x77");
    
    // 在屏幕上显示错误信息
    display.clearDisplay();
    display.setTextSize(1);
    display.setTextColor(SSD1306_WHITE);
    display.setCursor(0, 0);
    display.println("iCEasy");
    display.println("");
    display.println("BMP280 Error!");
    display.println("Check wiring");
    display.display();
    
    for(;;); // 死循环
  }
  Serial.println("BMP280 sensor initialized successfully");
  
  // 配置BMP280采样参数
  bmp.setSampling(Adafruit_BMP280::MODE_NORMAL,     // 正常工作模式
                  Adafruit_BMP280::SAMPLING_X2,     // 温度过采样 x2
                  Adafruit_BMP280::SAMPLING_X16,    // 气压过采样 x16
                  Adafruit_BMP280::FILTER_X16,      // 滤波 x16
                  Adafruit_BMP280::STANDBY_MS_500); // 待机时间 500ms
  
  Serial.println("\n========================================");
  Serial.println("System ready! Starting data display...");
  Serial.println("========================================\n");
  
  delay(500);
}

void loop() {
  // 读取BMP280传感器数据
  float temperature = bmp.readTemperature();      // 温度 (°C)
  float pressure = bmp.readPressure() / 100.0;    // 气压 (hPa)
  float altitude = bmp.readAltitude(SEALEVELPRESSURE_HPA); // 海拔 (m)
  
  // 串口输出数据
  Serial.println("--- BMP280 Data ---");
  Serial.print("Temperature: ");
  Serial.print(temperature);
  Serial.println(" °C");
  Serial.print("Pressure:    ");
  Serial.print(pressure);
  Serial.println(" hPa");
  Serial.print("Altitude:    ");
  Serial.print(altitude);
  Serial.println(" m");
  Serial.println();
  
  // 清空显示缓冲区
  display.clearDisplay();
  
  // ========== 第一行：iCEasy ==========
  display.setTextSize(2);
  display.setTextColor(SSD1306_WHITE);
  display.setCursor(0, 0);
  display.print("iCEasy");
  
  // ========== 第二行：温度 ==========
  display.setTextSize(1);
  display.setCursor(0, 18);
  display.print("Temp: ");
  display.print(temperature, 1);  // 保留1位小数
  display.print(" C");
  
  // ========== 第三行：气压 ==========
  display.setCursor(0, 32);
  display.print("Pres: ");
  display.print(pressure, 1);     // 保留1位小数
  display.print(" hPa");
  
  // ========== 第四行：海拔 ==========
  display.setCursor(0, 46);
  display.print("Alti: ");
  display.print(altitude, 1);     // 保留1位小数
  display.print(" m");
  
  // 发送到显示屏
  display.display();
  
  // 每2秒更新一次
  delay(2000);
}

